

The General Triangle Is Unique

Joachim Breitner

April 1, 2011

Some acute-angled triangles are special, e.g. right-angled or isosceles triangles. Some are not of this kind, but, without measuring angles, look as if they are. In that sense, there is exactly one general triangle. This well-known fact[1] is proven here formally.

theory *GeneralTriangle*
imports *Complex-Main*
begin

1 Type definitions

Since we are only considering acute-angled triangles, we define *angles* as numbers from the real interval $[0 \ldots 90]$.

abbreviation $angles \equiv \{ x::real .\ 0 \leq x \wedge x \leq 90 \}$

Triangles are represented as lists consisting of exactly three angles which add up to 180°. As we consider triangles up to similarity, we assume the angles to be given in ascending order.

Isabelle expects us to prove that the type is not empty, which we do by an example.

definition
triangle =
{ *l* . *l* ∈ *lists angles* ∧
length l = *3* ∧
sum-list l = *180* ∧

```
        sorted l
        }

typedef triangle = triangle
  unfolding triangle-def
  apply (rule-tac x = [45,45,90] in exI)
  apply auto
  done
```

For convenience, the following lemma gives us easy access to the three angles of a triangle and their properties.

```
lemma unfold-triangle:
  obtains a b c
  where Rep-triangle t = [a,b,c]
    and a ∈ angles
    and b ∈ angles
    and c ∈ angles
    and a + b + c = 180
    and a ≤ b
    and b ≤ c
proof−
  obtain a b c where
  a = Rep-triangle t ! 0 and b = Rep-triangle t ! 1 and c = Rep-triangle t ! 2
    using Rep-triangle[of t]
    by (auto simp add:triangle-def)
  hence Rep-triangle t = [a,b,c]
    using Rep-triangle[of t]
    apply (auto simp add:triangle-def)
    apply (cases Rep-triangle t, auto)
    apply (case-tac list, auto)
    apply (case-tac lista, auto)
    done
  with that show thesis
    using Rep-triangle[of t]
    by (auto simp add:triangle-def)
qed
```

2 Property definitions

Two angles can be considered too similar if they differ by less than 15°. This number is obtained heuristically by a field experiment with an 11th grade class and was chosen that statistically, 99% will consider the angles as different.

```
definition similar-angle :: real ⇒ real ⇒ bool (infix ‹∼› 50)
  where similar-angle x y = (abs (x − y) < 15)
```

The usual definitions of right-angled and isosceles, using the just introduced similarity for comparison of angles.

```
definition right-angled
```

where *right-angled l* = (∃ *x* ∈ *set* (*Rep-triangle l*). *x* ∼ *90*)

definition *isosceles*
where *isosceles l* = ((*Rep-triangle l*) ! *0* ∼ (*Rep-triangle l*) ! *1* ∨
(*Rep-triangle l*) ! *1* ∼ (*Rep-triangle l*) ! (*Suc 1*))

A triangle is special if it is isosceles or right-angled, and general if not. Equilateral triangle are isosceles and thus not mentioned on their own here.

definition *special*
where *special t* = (*isosceles t* ∨ *right-angled t*)

definition *general*
where *general t* = (¬ *special t*)

3 The Theorem

theorem ∃! *t*. *general t*

The proof proceeds in two steps: There is a general triangle, and it is unique. For the first step we give the triangle (angles 45°, 60°and 75°), show that it is a triangle and that it is general.

proof
have *is-t* [*simp*]: [*45*, *60*, *75*] ∈ *triangle* **by** (*auto simp add*: *triangle-def*)
show *general* (*Abs-triangle* [*45*,*60*,*75*]) (**is** *general ?t*)
by (*auto simp add*:*general-def special-def isosceles-def right-angled-def*
Abs-triangle-inverse similar-angle-def)
next

For the second step, we give names to the three angles and successively find upper bounds to them.

fix *t*
obtain *a b c* **where**
abc: *Rep-triangle t* = [*a*,*b*,*c*]
and *a* ∈ *angles* **and** *b* ∈ *angles* **and** *c* ∈ *angles*
and *a* ≤ *b* **and** *b* ≤ *c*
and *a* + *b* + *c* = *180*
by (*rule unfold-triangle*)

assume *general t*
hence *ni*: ¬ *isosceles t* **and** *nra*: ¬ *right-angled t*
by (*auto simp add*: *general-def special-def*)

have ¬ *c* ∼ *90* **using** *nra abc*
by (*auto simp add*:*right-angled-def*)
hence *c* ≤ *75* **using** ‹*c* ∈ *angles*›
by (*auto simp add*:*similar-angle-def*)

have ¬ *b* ∼ *c* **using** *ni abc*
by (*auto simp add*:*isosceles-def*)
hence *b* ≤ *60* **using** ‹*b* ≤ *c*› **and** ‹*c* ≤ *75*›

by (*auto simp add*:*similar-angle-def*)

have $\neg\ a \sim b$ **using** *ni abc*
by (*auto simp add*:*isosceles-def*)
hence $a \leq 45$ **using** ‹$a \leq b$› **and** ‹$b \leq 60$›
by (*auto simp add*:*similar-angle-def*)

The upper bound is actually the value, or we would not have a triangle

have $\neg\ (c < 75 \vee b < 60 \vee a < 45)$
proof
assume $c < 75 \vee b < 60 \vee a < 45$
hence $a + b + c < 180$ **using** ‹$c \leq 75$› ‹$b \leq 60$› ‹$a \leq 45$›
and ‹$a \in angles$› ‹$b \in angles$› ‹$c \in angles$›
by *auto*
thus *False* **using** ‹$a + b + c = 180$› **by** *auto*
qed
hence $c = 75$ **and** $b = 60$ **and** $a = 45$
using ‹$c \leq 75$› ‹$b \leq 60$› ‹$a \leq 45$›
by *auto*

And this concludes the proof.

hence *Abs-triangle* (*Rep-triangle t*) = *Abs-triangle* [*45*, *60*, *75*]
using *abc* **by** *simp*
thus *t* = *Abs-triangle* [*45*, *60*, *75*] **by** (*simp add*: *Rep-triangle-inverse*)
qed

end

References

[1] B. Tergan. Das allgemeine Dreieck. *Praxis der Mathematik*, 23:48–51, 1981.